In[246]:=

```
SetDirectory[NotebookDirectory[]];
<< MaTeX`
texStyle = {};
SetOptions[MaTeX,
  "BasePreamble" → {"\\usepackage{amsmath}", "\\usepackage{xcolor}",
    "\\usepackage{fourier}", "\\usepackage{ebgaramond}"}, FontSize → 11];
```

In[250]:=

```
αλ[λ_, μ_] := μ / (λ + 2 μ);
λα[α_, μ_] := μ (1 / α - 2);
CWeyl[d_, λ_, μ_] :=
  ((λ + 2 μ) ^ (-d / 2) + (d - 1) μ ^ (-d / 2)) / (4 Pi) ^ (d / 2) / Gamma[1 + d / 2];
BDirLiu[d_, λ_, μ_] :=
  - ((d - 1) / μ ^ ((d - 1) / 2) + 1 / (λ + 2 μ) ^ ((d - 1) / 2)) / 4 / (4 Pi) ^ ((d - 1) / 2) /
   Gamma[1 + (d - 1) / 2];
BDir[d_, α_, μ_] := -μ ^ ((1 - d) / 2) / (2 ^ (d + 1) Pi ^ ((d - 1) / 2) Gamma[(1 + d) / 2])
   (4 (d - 1) / Pi NIntegrate[τ ^ (d - 2) ArcTan[Sqrt[(1 - α τ ^ (-2)) (τ ^ (-2) - 1)]],
       {τ, Sqrt[α], 1}] + α ^ ((d - 1) / 2) + d - 1);
BDirSaVa2[α_, μ_] := 1 / (4 Pi Sqrt[μ]) (-1 - Sqrt[α] -
     4 / Pi NIntegrate[ArcTan[Sqrt[(1 - α / ξ^2) (1 / ξ^2 - 1)]], {ξ, Sqrt[α], 1}]);
BDirSaVa3[λ_, μ_] := -1 / (16 Pi) (3 λ^2 + 13 λ μ + 16 μ^2) / (λ^2 μ + 5 λ μ^2 + 6 μ^3);
γR[α_] := Sqrt[Min[x /. Solve[x^3 - 8 x^2 + 16 (-1 + α) + 8 (3 - 2 α) x == 0, x, Reals]]];
Bfree[d_, α_, μ_] := μ ^ ((1 - d) / 2) / (2 ^ (d + 1) Pi ^ ((d - 1) / 2) Gamma[(1 + d) / 2])
   (4 (d - 1) / Pi NIntegrate[
       τ ^ (d - 2) ArcTan[(τ ^ (-2) - 2) ^2 / (4 Sqrt[(1 - α τ ^ (-2)) (τ ^ (-2) - 1)])],
       {τ, Sqrt[α], 1}] + α ^ ((d - 1) / 2) + d - 5 + 4 γR[α] ^ (1 - d));
BfreeSaVa2[α_, μ_] :=
  1 / (4 Pi Sqrt[μ]) (4 / γR[α] - 3 + Sqrt[α] + 4 / Pi NIntegrate[ArcTan[
        (2 - 1 / ξ^2) ^2 / (4 Sqrt[(1 - α / ξ^2) (1 / ξ^2 - 1)])], {ξ, Sqrt[α], 1}]);
BfreeSaVa3[λ_, μ_] :=
  1 / (16 Pi) (3 (λ + 2 μ) ^2 - 3 (λ + 2 μ) μ + 2 μ^2) / ((λ + 2 μ) μ (λ + μ));
```

In[261]:=

```
Clear[findAllRoots]
SyntaxInformation[findAllRoots] = {"LocalVariables" → {"Plot", {2, 2}},
   "ArgumentsPattern" → {_, _, OptionsPattern[]}};
SetAttributes[findAllRoots, HoldAll];

Options[findAllRoots] = Join[{"ShowPlot" → False, PlotRange → All},
   FilterRules[Options[Plot], Except[PlotRange]]];

findAllRoots[fn_, {l_, lmin_, lmax_}, opts : OptionsPattern[]] :=
 Module[{pl, p, x, localFunction, brackets},
  localFunction = ReleaseHold[Hold[fn] /. HoldPattern[l] :→ x];
  If[lmin ≠ lmax, pl = Plot[localFunction, {x, lmin, lmax}, Evaluate@
      FilterRules[Join[{opts}, Options[findAllRoots]], Options[Plot]]];
   p = Cases[pl, Line[{x__}] :→ x, Infinity];
   If[OptionValue["ShowPlot"],
    Print[Show[pl, PlotLabel → "Finding roots for this function",
      ImageSize → 200, BaseStyle → {FontSize → 8}]]], p = {}];
  brackets =
   Map[First, Select[(*This Split trick pretends that two points on the
        curve are "equal" if the function values have _opposite _ sign.Pairs
        of such sign-changes form the brackets for the subsequent FindRoot*)
     Split[p, Sign[Last[#2]] == -Sign[Last[#1]] &], Length[#1] == 2 &], {2}];
  x /. Apply[FindRoot[localFunction == 0, {x, ##1}] &, brackets, {1}] /. x → {}]
```

In[274]:=

```
ClearAll[ℒ];
ℒ[V_, λ_, μ_] := μ Curl[Curl[V, {x1, x2, x3}], {x1, x2, x3}] -
  (λ + 2 μ) Grad[Div[V, {x1, x2, x3}], {x1, x2, x3}];
```

In[275]:=

```
ℒ[{v1[x1, x2, x3], v2[x1, x2, x3], v3[x1, x2, x3]}, λ, μ]
```

Out[275]=

$$\Big\{\mu\left(-v_1^{(0,0,2)}[x_1,x_2,x_3]-v_1^{(0,2,0)}[x_1,x_2,x_3]+v_3^{(1,0,1)}[x_1,x_2,x_3]+v_2^{(1,1,0)}[x_1,x_2,x_3]\right)-(\lambda+2\mu)\left(v_3^{(1,0,1)}[x_1,x_2,x_3]+v_2^{(1,1,0)}[x_1,x_2,x_3]+v_1^{(2,0,0)}[x_1,x_2,x_3]\right),$$
$$-\left((\lambda+2\mu)\left(v_3^{(0,1,1)}[x_1,x_2,x_3]+v_2^{(0,2,0)}[x_1,x_2,x_3]+v_1^{(1,1,0)}[x_1,x_2,x_3]\right)\right)+\mu\left(-v_2^{(0,0,2)}[x_1,x_2,x_3]+v_3^{(0,1,1)}[x_1,x_2,x_3]+v_1^{(1,1,0)}[x_1,x_2,x_3]-v_2^{(2,0,0)}[x_1,x_2,x_3]\right),$$
$$-\left((\lambda+2\mu)\left(v_3^{(0,0,2)}[x_1,x_2,x_3]+v_2^{(0,1,1)}[x_1,x_2,x_3]+v_1^{(1,0,1)}[x_1,x_2,x_3]\right)\right)+\mu\left(v_2^{(0,1,1)}[x_1,x_2,x_3]-v_3^{(0,2,0)}[x_1,x_2,x_3]+v_1^{(1,0,1)}[x_1,x_2,x_3]-v_3^{(2,0,0)}[x_1,x_2,x_3]\right)\Big\}$$

In[276]:=

```
1 / (λ + 2 μ) ℒ[Grad[ψ[x1, x2, x3], {x1, x2, x3}], λ, μ] +
 Laplacian[Grad[ψ[x1, x2, x3], {x1, x2, x3}], {x1, x2, x3}]
1 / μ ℒ[Curl[{0, 0, ψ[x1, x2, x3]}, {x1, x2, x3}], λ, μ] +
 Laplacian[Curl[{0, 0, ψ[x1, x2, x3]}, {x1, x2, x3}], {x1, x2, x3}]
1 / μ ℒ[Curl[Curl[{0, 0, ψ[x1, x2, x3]}, {x1, x2, x3}], {x1, x2, x3}], λ, μ] +
 Laplacian[Curl[Curl[{0, 0, ψ[x1, x2, x3]}, {x1, x2, x3}], {x1, x2, x3}], {x1, x2, x3}]
```

Out[276]=

```
{0, 0, 0}
```

Out[277]=

```
{0, 0, 0}
```

Out[278]=

```
{0, 0, 0}
```

In[279]:=

```
ClearAll[κ];
-Laplacian[Exp[I (ξ1 x1 + ξ2 x2 - Sqrt[κ - ξ1^2 - ξ2^2] x3)], {x1, x2, x3}] /
  Exp[I (ξ1 x1 + ξ2 x2 - Sqrt[κ - ξ1^2 - ξ2^2] x3)] // Simplify
```

Out[279]=

```
κ
```

In[280]:=

```
V = Grad[c1 Exp[I (ξ1 x1 + ξ2 x2 + Sqrt[Λ / (λ + 2 μ) - ξ1^2 - ξ2^2] x3)]  +
    c2 Exp[I (ξ1 x1 + ξ2 x2 - Sqrt[Λ / (λ + 2 μ) - ξ1^2 - ξ2^2] x3)], {x1, x2, x3}];
W = Curl[{0, 0, c3 Exp[I (ξ1 x1 + ξ2 x2 + Sqrt[Λ / μ - ξ1^2 - ξ2^2] x3)] +
     c4 Exp[I (ξ1 x1 + ξ2 x2 - Sqrt[Λ / μ - ξ1^2 - ξ2^2] x3)]}, {x1, x2, x3}];
Y = Curl[Curl[{0, 0, c5 Exp[I (ξ1 x1 + ξ2 x2 + Sqrt[Λ / μ - ξ1^2 - ξ2^2] x3)] + c6
       Exp[I (ξ1 x1 + ξ2 x2 - Sqrt[Λ / μ - ξ1^2 - ξ2^2] x3)]}, {x1, x2, x3}], {x1, x2, x3}];
ℒ[V, λ, μ] - Λ V // Simplify
ℒ[W, λ, μ] - Λ W // Simplify
ℒ[Y, λ, μ] - Λ Y // Simplify
```

Out[283]=

```
{0, 0, 0}
```

Out[284]=

```
{0, 0, 0}
```

Out[285]=

```
{0, 0, 0}
```

In[286]:=

```
U = V + W + Y
```

Out[286]=

$$\Big\{\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\right)} c_1 \xi_1+\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\right)} c_2 \xi_1+$$
$$\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_3 \xi_2+\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_4 \xi_2-$$
$$e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_5 \xi_1 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}+e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_6 \xi_1 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2},$$
$$-\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_3 \xi_1-\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_4 \xi_1+$$
$$\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\right)} c_1 \xi_2+\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\right)} c_2 \xi_2-$$
$$e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_5 \xi_2 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}+e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_6 \xi_2 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2},$$
$$e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_5 \xi_1^2+e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_6 \xi_1^2+e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_5 \xi_2^2+$$
$$e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\mu}-\xi_1^2-\xi_2^2}\right)} c_6 \xi_2^2+\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2+x_3 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\right)} c_1 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}-$$
$$\mathrm{i}\, e^{\mathrm{i}\left(x_1 \xi_1+x_2 \xi_2-x_3 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\right)} c_2 \sqrt{\frac{\Lambda}{\lambda+2\mu}-\xi_1^2-\xi_2^2}\Big\}$$

In[287]:=

```
U0 = (U /. {ξ1 → 0, ξ2 → 0})
CoefficientArrays[{U0⟦3⟧ /. x3 → 0, U0⟦3⟧ /. x3 → h}, {c1, c2}]⟦2⟧ // Normal //
  Det // FullSimplify
```

Out[287]=

$$\left\{0,\ 0,\ \mathrm{i}\, e^{\mathrm{i}\sqrt{\frac{\Lambda}{\lambda+2\mu}}\, x_3} \sqrt{\frac{\Lambda}{\lambda+2\mu}}\, c_1-\mathrm{i}\, e^{-\mathrm{i}\sqrt{\frac{\Lambda}{\lambda+2\mu}}\, x_3} \sqrt{\frac{\Lambda}{\lambda+2\mu}}\, c_2\right\}$$

Out[288]=

$$-\frac{2\, \mathrm{i}\, \Lambda \operatorname{Sin}\left[h \sqrt{\frac{\Lambda}{\lambda+2\mu}}\right]}{\lambda+2\mu}$$

In[289]:=

```
f = ((CoefficientArrays[{(Exp[-I (ξ1 x1 + ξ2 x2)] U /. x3 → 0),
        (Exp[-I (ξ1 x1 + ξ2 x2)] U /. x3 → h)} // Flatten // Simplify,
      Table[cj, {j, 1, 6}]][[2]] // Normal // Det // Simplify) /.
    {ξ1 → m1, ξ2 → m2, m1^2 + m2^2 → R, - m1^2 - m2^2 → -R} ) // FullSimplify;
f1 = f /. {m1^2 + m2^2 → R, - m1^2 - m2^2 → -R} // Simplify;
mydet1 =
```

$$f1 \;/.\; \left\{\sqrt{-R + \frac{\Lambda}{\lambda + 2\,\mu}} \to \text{Sqrt[C1]},\ \frac{\sqrt{\Lambda - R\,\mu}}{\sqrt{\mu}} \to \text{Sqrt[C2]},\ \sqrt{\Lambda - R\,\mu} \to \sqrt{C2}\ \text{Sqrt}[\mu]\right\} \;//$$

```
  FullSimplify
```

Out[291]=

$$8\,i\,R^2\,\text{Sin}\left[\sqrt{C2}\;h\right]\left(2\sqrt{C1}\sqrt{C2}\,R\left(-1+\text{Cos}\left[\sqrt{C1}\;h\right]\text{Cos}\left[\sqrt{C2}\;h\right]\right) - \frac{\left(\Lambda^2 + 2\,R^2\,\mu\,(\lambda + 2\,\mu) - R\,\Lambda\,(\lambda + 3\,\mu)\right)\text{Sin}\left[\sqrt{C1}\;h\right]\text{Sin}\left[\sqrt{C2}\;h\right]}{\mu\,(\lambda + 2\,\mu)}\right)$$

In[292]:=

```
(* Case Λ< R μ    *)
(mydet1 /. {Sqrt[C1] → I Sqrt[Abs[C1]], Sqrt[C2] → I Sqrt[Abs[C2]]}) //
 Simplify
```

Out[292]=

$$-8\,R^2\,\text{Sinh}\left[h\sqrt{\text{Abs}[C2]}\right]\left(-2\,R\sqrt{\text{Abs}[C1]}\sqrt{\text{Abs}[C2]}\left(-1+\text{Cosh}\left[h\sqrt{\text{Abs}[C1]}\right]\text{Cosh}\left[h\sqrt{\text{Abs}[C2]}\right]\right) + \frac{\left(\Lambda^2 + 2\,R^2\,\mu\,(\lambda + 2\,\mu) - R\,\Lambda\,(\lambda + 3\,\mu)\right)\text{Sinh}\left[h\sqrt{\text{Abs}[C1]}\right]\text{Sinh}\left[h\sqrt{\text{Abs}[C2]}\right]}{\mu\,(\lambda + 2\,\mu)}\right)$$

In[293]:=

```
(* Case   R μ ≤ Λ < R (λ+2 μ) *)
(mydet1 /. {Sqrt[C1] → I Sqrt[Abs[C1]]}) // Simplify
```

Out[293]=

$$8\,R^2\,\text{Sin}\left[\sqrt{C2}\;h\right]\left(-2\sqrt{C2}\,R\sqrt{\text{Abs}[C1]}\left(-1+\text{Cos}\left[\sqrt{C2}\;h\right]\text{Cosh}\left[h\sqrt{\text{Abs}[C1]}\right]\right) + \frac{\left(\Lambda^2 + 2\,R^2\,\mu\,(\lambda + 2\,\mu) - R\,\Lambda\,(\lambda + 3\,\mu)\right)\text{Sin}\left[\sqrt{C2}\;h\right]\text{Sinh}\left[h\sqrt{\text{Abs}[C1]}\right]}{\mu\,(\lambda + 2\,\mu)}\right)$$

In[294]:=

```
MyDetSign1[λ_, μ_, h_, R_ , Λ_] :=
  (* for Λ < R μ *) With[{AC1 = R - Λ/(λ + 2 μ), AC2 = -Λ / μ + R},
   (2 R Sqrt[AC1] Sqrt[AC2] (-1 + Cosh[h Sqrt[AC1]] Cosh[h Sqrt[AC2]]) -
      ((Λ^2 + 2 R^2 μ (λ + 2 μ) - R Λ (λ + 3 μ)) Sinh[h Sqrt[AC1]] Sinh[h Sqrt[AC2]])/(μ (λ + 2 μ))) /
    (Cosh[h Sqrt[AC1]] Cosh[h Sqrt[AC2]])];
MyDetSign2Part1[λ_, μ_, h_, R_ , Λ_] :=  (* for R μ < Λ < R(λ+2 μ)   *)
  With[{AC1 = R - Λ/(λ + 2 μ), AC2 = Λ / μ - R}, Sin[Sqrt[AC2] h]];
MyDetSign2Part2[λ_, μ_, h_, R_ , Λ_] :=  (* for R μ < Λ < R(λ+2 μ)   *)
  With[{AC1 = R - Λ/(λ + 2 μ), AC2 = Λ / μ - R}, (-2 (-1 + Cosh[h Sqrt[AC1]] Cos[h Sqrt[AC2]]) +
      ((Λ^2 + 2 R^2 μ (λ + 2 μ) - R Λ (λ + 3 μ)) Sinh[h Sqrt[AC1]] Sin[h Sqrt[AC2]])/(μ (λ + 2 μ) R Sqrt[AC1] Sqrt[AC2])) / Cosh[h
      Sqrt[AC1]]];
MyDetSign3Part1[λ_, μ_, h_, R_ , Λ_] :=  (* for Λ > R(λ+2 μ)   *)
  With[{C1 = -R + Λ/(λ + 2 μ), C2 = Λ / μ - R}, 8 Sin[Sqrt[C2] h]];
MyDetSign3Part2[λ_, μ_, h_, R_ , Λ_] :=
  (* for Λ > R(λ+2 μ)   *) With[{C1 = -R + Λ/(λ + 2 μ), C2 = Λ / μ - R},
   (2 (-1 + Cos[Sqrt[C1] h] Cos[Sqrt[C2] h]) -
     ((Λ^2 + 2 R^2 μ (λ + 2 μ) - R Λ (λ + 3 μ)) Sin[Sqrt[C1] h] Sin[Sqrt[C2] h])/(μ (λ + 2 μ) Sqrt[C1] Sqrt[C2] R))];
```

In[299]:=

```
(*Example:λ=0,μ=1,h=4,R=9 *)
Plot[MyDetSign1[0, 1, 4, 9, Λ], {Λ, 0, 9}]
Show[Plot[{MyDetSign2Part1[0, 1, 4, 9, Λ],
   MyDetSign2Part2[0, 1, 4, 9, Λ]}, {Λ, 9, 18}],
 ListPlot[
  Thread[{Λ /. NSolve[MyDetSign2Part2[0, 1, 4, 9, Λ] == 0 && 9 < Λ < 18], 0}],
  PlotStyle → {{Orange, PointSize[Medium]}}]]
Show[Plot[{MyDetSign3Part1[0, 1, 4, 9, Λ],
   MyDetSign3Part2[0, 1, 4, 9, Λ]}, {Λ, 18, 100}],
 ListPlot[
  Thread[{Λ /. NSolve[MyDetSign3Part2[0, 1, 4, 9, Λ] == 0 && 18 < Λ < 100], 0}],
  PlotStyle → {{Orange, PointSize[Medium]}}]]
```

Out[299]=



Out[300]=



Out[301]=



In[302]:=

```
(* numerically computed eigenvalues *)
MyEigs[λ_, μ_, h_, ΛMax_] :=
  Module[{evs, R, Rmax, n, Rsq, solns21, solns22, solns31, solns32, j},
   evs = Table[{0, 1, N[(λ + 2 μ) Pi^2 n^2 / h^2]},
     {n, 1, h Sqrt[ΛMax] / (Pi Sqrt[λ + 2 μ])}];
   (* the eigenvalues above come from R=0, with multiplicity one *)
   Do[Rsq = SquaresR[2, R];
    If[Rsq ≠ 0,
     solns21 = Λ /. NSolve[
        MyDetSign2Part1[λ, μ, h, R, Λ] == 0 && μ R < Λ < (λ + 2 μ) R && Λ ≤ ΛMax, Λ];
     solns22 = Λ /. NSolve[
        MyDetSign2Part2[λ, μ, h, R, Λ] == 0 && μ R < Λ < (λ + 2 μ) R && Λ ≤ ΛMax, Λ];
     Do[AppendTo[evs, {R, Rsq, solns21[[j]]}], {j, 1, Length[solns21]}];
     Do[AppendTo[evs, {R, Rsq, solns22[[j]]}], {j, 1, Length[solns22]}];
     solns31 =
      Λ /. NSolve[MyDetSign3Part1[λ, μ, h, R, Λ] == 0 && (λ + 2 μ) R < Λ ≤ ΛMax, Λ];
     solns32 =
      Λ /. NSolve[MyDetSign3Part2[λ, μ, h, R, Λ] == 0 && (λ + 2 μ) R < Λ ≤ ΛMax, Λ];
     Do[AppendTo[evs, {R, Rsq, solns31[[j]]}], {j, 1, Length[solns31]}];
     Do[AppendTo[evs, {R, Rsq, solns32[[j]]}], {j, 1, Length[solns32]}];
    ]
    , {R, 1, ΛMax / μ}];
   SortBy[evs, N[#[[3]]] &]
  ];
MyEigs3D2D[evs_] := Module[{evs2, j, Rsq2},
  evs2 = {};
  Do[Rsq2 = SquaresR[1, evs[[j]][[1]]];
   If[Rsq2 ≠ 0, AppendTo[evs2, {evs[[j]][[1]], Rsq2, evs[[j]][[3]]}]],
   {j, 1, Length[evs]}];
  evs2
 ]
```

In[304]:=

```
λs = {-1 / 2, 3, 100}; hs = {1 / 2, 2}; Lmax = {1000, 500, 250};
htext = {"0.5", "2"}; ylim = {-5000, -2500, -1200};
lambdatext = {"-0.5", "3", "100"};
xticks = Table[
   {{Lmax⟦j⟧ / 2, MaTeX[Lmax⟦j⟧ / 2]}, {Lmax⟦j⟧, MaTeX[Lmax⟦j⟧]}}, {j, 1, 3}] ;
yticks = Table[
   {{ylim⟦i⟧ / 2, MaTeX[ylim⟦i⟧ / 2]}, {ylim⟦i⟧, MaTeX[ylim⟦i⟧]}}, {i, 1, 2}] ;
```

In[308]:=

```
EVs = Table[MyEigs[λs⟦j⟧, 1, hs⟦i⟧, Lmax⟦i⟧], {i, 1, 2}, {j, 1, 3}];
```

In[309]:=

```
MyNlambda3d[evs_, Λ_] :=
  Sum[evs⟦j⟧⟦2⟧ UnitStep[Λ - evs⟦j⟧⟦3⟧], {j, 1, Length[evs]}];
Vol3[h_] := (2 Pi) ^2 h;
Area3[h_] := 2 (2 Pi) ^2;
```

In[312]:=

```
graphicstableDir =
 Table[
  Plot[{MyNlambda3d[EVs⟦i⟧⟦j⟧, Λ] - CWeyl[3, λs⟦j⟧, 1] × Vol3[hs⟦i⟧] Λ^(3 / 2),
    BDir[3, αλ[λs⟦j⟧, 1], 1] × Area3[hs⟦i⟧] Λ,
    BDirLiu[3, λs⟦j⟧, 1] × Area3[hs⟦i⟧] Λ},
   {Λ, 0, Lmax⟦i⟧}, PlotStyle → {Black, Blue, Red},
   Axes → True,
   AxesLabel → {MaTeX["\\Lambda"], None},
   PlotRange → {{0, Lmax⟦i⟧}, {ylim⟦i⟧, 0}},
   Ticks → {xticks⟦i⟧, yticks⟦i⟧},
   Epilog → Inset[
     Framed[MaTeX["\\lambda=" <> lambdatext⟦j⟧ <> ",\\quad h=" <> htext⟦i⟧],
      RoundingRadius → 2, ContentPadding → False,
      FrameMargins → Tiny], Scaled[{0.1, 0}], Scaled[{0, 0}]]
  ], {i, 1, 2}, {j, 1, 3}]
(* plotsCylinder3=
 GraphicsGrid[graphicstableDir, ImageSize→Full, Frame→True, Dividers→All] *)
```

Out[312]=



In[313]:=

```
legencyl = LineLegend[{Black, Blue, Red},
  MaTeX[{"\\text{computed }\\mathcal{N}_{\\mathrm{Dir}}(\\Lambda)-
     a (2\\pi)^2 h \\Lambda^{3/2}\\qquad\\qquad",
    "8\\pi^2 b_{\\mathrm{Dir}}\\Lambda\\qquad\\qquad",
    "8\\pi^2 b_{\\mathrm{Dir}}^{\\mathrm{Liu}}\\Lambda"}],
  LegendFunction → None, LegendLayout → {"Row", 1}]
```

Out[313]=

— computed $\mathcal{N}_{\mathrm{Dir}}(\Lambda) - a(2\pi)^2 h\Lambda^{3/2}$ — $8\pi^2 b_{\mathrm{Dir}}\Lambda$ — $8\pi^2 b_{\mathrm{Dir}}^{\mathrm{Liu}}\Lambda$

In[314]:=

```
figcyl = GraphicsColumn[{GraphicsRow[graphicstableDir[[1]],
     ImageSize → Full, Spacings → {Scaled[0.05], Automatic}],
    GraphicsRow[graphicstableDir[[2]],
     ImageSize → Full, Spacings → {Scaled[0.05], Automatic}],
    legencyl}, ImageSize → Full, Spacings → {0, Automatic}]
```

Out[314]=



In[∘]:= `Export["figcyl.pdf", figcyl]`

Out[∘]=

```
figcyl.pdf
```

In[315]:=

```
TU = {μ (D[U[[1]], x3] + D[U[[3]], x1]), μ (D[U[[2]], x3] + D[U[[3]], x2]),
     λ (D[U[[1]], x1] + D[U[[2]], x2] + D[U[[3]], x3]) + 2 μ D[U[[3]], x3]}
    Exp[-I (ξ1 x1 + ξ2 x2)] // Simplify;
```

In[316]:=

```
TU0 = {μ (D[U0〚1〛, x3] + D[U0〚3〛, x1]), μ (D[U0〚2〛, x3] + D[U0〚3〛, x2]),
    λ (D[U0〚1〛, x1] + D[U0〚2〛, x2] + D[U0〚3〛, x3]) + 2 μ D[U0〚3〛, x3]} // Simplify;
CoefficientArrays[{TU0〚3〛 /. x3 → 0, TU0〚3〛 /. x3 → h}, {c1, c2}]〚2〛 // Normal //
  Det // FullSimplify
```

Out[317]=

$$-2\,i\,\Lambda^2\,\mathrm{Sin}\left[h\sqrt{\frac{\Lambda}{\lambda+2\mu}}\right]$$

In[318]:=

```
Tfmatrix =
 (((CoefficientArrays[{(TU /. x3 → 0), (TU /. x3 → h)} // Flatten // Simplify,
         Table[cj, {j, 1, 6}]]〚2〛 // Normal) /.
      {ξ1 → m1, ξ2 → m2, m2^2 → R - m1^2, -m2^2 → -R + m1^2}) // Simplify) /.
   {ξ1 → m1, ξ2 → m2, m2^2 → R - m1^2, -m2^2 → -R + m1^2} // Simplify
```

Out[318]=

$$\Bigg\{\Bigg\{-2\,m1\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},\ 2\,m1\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},$$
$$-m2\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ m2\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ -i\,m1\,(\Lambda-2R\mu),\ -i\,m1\,(\Lambda-2R\mu)\Bigg\},$$
$$\Bigg\{-2\,m2\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},\ 2\,m2\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},\ m1\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,$$
$$-m1\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ -i\,m2\,(\Lambda-2R\mu),\ -i\,m2\,(\Lambda-2R\mu)\Bigg\},$$
$$\Bigg\{-\Lambda+2R\mu,\ -\Lambda+2R\mu,\ 0,\ 0,\ 2\,i\,R\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ -2\,i\,R\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu\Bigg\},$$
$$\Bigg\{-2\,e^{i h\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}}\,m1\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},\ 2\,e^{-i h\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}}\,m1\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},$$
$$-e^{i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m2\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ e^{-i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m2\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,$$
$$-i\,e^{i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m1\,(\Lambda-2R\mu),\ -i\,e^{-i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m1\,(\Lambda-2R\mu)\Bigg\},$$
$$\Bigg\{-2\,e^{i h\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}}\,m2\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},\ 2\,e^{-i h\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}}\,m2\,\mu\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}},$$
$$e^{i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m1\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ -e^{-i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m1\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ -i\,e^{i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m2\,(\Lambda-2R\mu),$$
$$-i\,e^{-i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,m2\,(\Lambda-2R\mu)\Bigg\},\ \Bigg\{e^{i h\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}}\,(-\Lambda+2R\mu),\ -e^{-i h\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}}\,(\Lambda-2R\mu),$$
$$0,\ 0,\ 2\,i\,e^{i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,R\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu,\ -2\,i\,e^{-i h\sqrt{-R+\frac{\Lambda}{\mu}}}\,R\sqrt{-R+\frac{\Lambda}{\mu}}\,\mu\Bigg\}\Bigg\}$$

In[319]:=

```
Tmatrix1 =
 (((Tfmatrix /. Λ → μ R) // FullSimplify) /. Sqrt[-R (λ + μ)/(λ + 2 μ)] → I Sqrt[R (λ + μ)/(λ + 2 μ)]) //
   Simplify;
Tmatrix1 // TableForm
Tmatrix1 // NullSpace
```

Out[320]//TableForm=

$-2\,i\,m1\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	$2\,i\,m1\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	0	0	$i\,m1\,R\,\mu$	$i\,m1\,R\,\mu$
$-2\,i\,m2\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	$2\,i\,m2\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	0	0	$i\,m2\,R\,\mu$	$i\,m2\,R\,\mu$
$R\,\mu$	$R\,\mu$	0	0	0	0
$-2\,i\,e^{-h\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}}\,m1\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	$2\,i\,e^{h\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}}\,m1\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	0	0	$i\,m1\,R\,\mu$	$i\,m1\,R\,\mu$
$-2\,i\,e^{-h\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}}\,m2\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	$2\,i\,e^{h\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}}\,m2\,\mu\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}$	0	0	$i\,m2\,R\,\mu$	$i\,m2\,R\,\mu$
$e^{-h\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}}\,R\,\mu$	$e^{h\sqrt{\frac{R(\lambda+\mu)}{\lambda+2\mu}}}\,R\,\mu$	0	0	0	0

Out[321]=

{{0, 0, 0, 0, -1, 1}, {0, 0, 0, 1, 0, 0}, {0, 0, 1, 0, 0, 0}}

In[322]:=

```
U /. {c1 → 0, c2 → 0, c3 → 0, c4 → 0, c5 → -1, c6 → 1, Λ/μ - ξ1^2 - ξ2^2 → 0} // Simplify
U1 =
  U /. {c1 → 0, c2 → 0, c3 → 1, c4 → 0, c5 → 0, c6 → 0, Λ/μ - ξ1^2 - ξ2^2 → 0} // FullSimplify
U /. {c1 → 0, c2 → 0, c3 → 0, c4 → 1, c5 → 0, c6 → 0, Λ/μ - ξ1^2 - ξ2^2 → 0} // FullSimplify
```

Out[322]=

{0, 0, 0}

Out[323]=

$\{i\,e^{i(x_1\xi_1+x_2\xi_2)}\,\xi_2,\ -i\,e^{i(x_1\xi_1+x_2\xi_2)}\,\xi_1,\ 0\}$

Out[324]=

$\{i\,e^{i(x_1\xi_1+x_2\xi_2)}\,\xi_2,\ -i\,e^{i(x_1\xi_1+x_2\xi_2)}\,\xi_1,\ 0\}$

In[325]:=

```
ℒ[U1, λ, μ][[1 ;; 2]] / U1[[1 ;; 2]] // Simplify
```

Out[325]=

$\{\mu\,(\xi_1^2+\xi_2^2),\ \mu\,(\xi_1^2+\xi_2^2)\}$

In[326]:=

```
U11 = ComplexExpand[U1 + Conjugate[U1]] / 2 // Factor
U12 = ComplexExpand[U1 - Conjugate[U1]] / 2 // Factor
```

Out[326]=

$\{-\text{Sin}[x_1\,\xi_1 + x_2\,\xi_2]\,\xi_2,\ \text{Sin}[x_1\,\xi_1 + x_2\,\xi_2]\,\xi_1,\ 0\}$

Out[327]=

$\{i\,\text{Cos}[x_1\,\xi_1 + x_2\,\xi_2]\,\xi_2,\ -i\,\text{Cos}[x_1\,\xi_1 + x_2\,\xi_2]\,\xi_1,\ 0\}$

In[328]:=

```
{U1 /. {ξ1 → 2, ξ2 → 0}, U1 /. {ξ1 → -2, ξ2 → 0},
  U1 /. {ξ2 → 2, ξ1 → 0}, U1 /. {ξ2 → -2, ξ1 → 0}} // Simplify
```

Out[328]=

$\{\{0, -2\,i\,e^{2\,i\,x_1}, 0\}, \{0, 2\,i\,e^{-2\,i\,x_1}, 0\}, \{2\,i\,e^{2\,i\,x_2}, 0, 0\}, \{-2\,i\,e^{-2\,i\,x_2}, 0, 0\}\}$

In[329]:=

```
{U1 /. {ξ1 → 2, ξ2 → 1}, U1 /. {ξ1 → -2, ξ2 → -1}, U1 /. {ξ1 → 2, ξ2 → -1},
   U1 /. {ξ1 → -2, ξ2 → 1}, U1 /. {ξ2 → 2, ξ1 → 1}, U1 /. {ξ2 → -2, ξ1 → -1},
   U1 /. {ξ2 → 2, ξ1 → -1}, U1 /. {ξ2 → -2, ξ1 → 1}} // Simplify // ComplexExpand
```

Out[329]=

$$\begin{aligned}
\{&\{i\,\text{Cos}[2\,x_1 + x_2] - \text{Sin}[2\,x_1 + x_2],\ -2\,i\,\text{Cos}[2\,x_1 + x_2] + 2\,\text{Sin}[2\,x_1 + x_2],\ 0\},\\
&\{-i\,\text{Cos}[2\,x_1 + x_2] - \text{Sin}[2\,x_1 + x_2],\ 2\,i\,\text{Cos}[2\,x_1 + x_2] + 2\,\text{Sin}[2\,x_1 + x_2],\ 0\},\\
&\{-i\,\text{Cos}[2\,x_1 - x_2] + \text{Sin}[2\,x_1 - x_2],\ -2\,i\,\text{Cos}[2\,x_1 - x_2] + 2\,\text{Sin}[2\,x_1 - x_2],\ 0\},\\
&\{i\,\text{Cos}[2\,x_1 - x_2] + \text{Sin}[2\,x_1 - x_2],\ 2\,i\,\text{Cos}[2\,x_1 - x_2] + 2\,\text{Sin}[2\,x_1 - x_2],\ 0\},\\
&\{2\,i\,\text{Cos}[x_1 + 2\,x_2] - 2\,\text{Sin}[x_1 + 2\,x_2],\ -i\,\text{Cos}[x_1 + 2\,x_2] + \text{Sin}[x_1 + 2\,x_2],\ 0\},\\
&\{-2\,i\,\text{Cos}[x_1 + 2\,x_2] - 2\,\text{Sin}[x_1 + 2\,x_2],\ i\,\text{Cos}[x_1 + 2\,x_2] + \text{Sin}[x_1 + 2\,x_2],\ 0\},\\
&\{2\,i\,\text{Cos}[x_1 - 2\,x_2] + 2\,\text{Sin}[x_1 - 2\,x_2],\ i\,\text{Cos}[x_1 - 2\,x_2] + \text{Sin}[x_1 - 2\,x_2],\ 0\},\\
&\{-2\,i\,\text{Cos}[x_1 - 2\,x_2] + 2\,\text{Sin}[x_1 - 2\,x_2],\ -i\,\text{Cos}[x_1 - 2\,x_2] + \text{Sin}[x_1 - 2\,x_2],\ 0\}\}
\end{aligned}$$

In[330]:=

```
Tmatrix2 = (((Tfmatrix /. Λ → (λ + 2 μ) R) // FullSimplify)) // Simplify;
Tmatrix2 // TableForm
Tmatrix2 // NullSpace
```

Out[331]//TableForm=

0	0	$-m2\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$m2\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$-i\,m1\,R\,\lambda$
0	0	$m1\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$-m1\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$-i\,m2\,R\,\lambda$
$-R\,\lambda$	$-R\,\lambda$	0	0	$2\,i\,R\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$
0	0	$-e^{i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,m2\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$e^{-i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,m2\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$-i\,e^{i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,m1\,R$
0	0	$e^{i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,m1\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$-e^{-i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,m1\,\mu\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}$	$-i\,e^{i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,m2\,R$
$-R\,\lambda$	$-R\,\lambda$	0	0	$2\,i\,e^{i\,h\,\sqrt{\frac{R\,(\lambda+\mu)}{\mu}}}\,R\,\mu$

Out[332]=

{{-1, 1, 0, 0, 0, 0}}

In[333]:=

```
U /. {c1 → -1, c2 → 1, c3 → 0, c4 → 0, c5 → 0, c6 → 0, Λ → (λ + 2 μ) (ξ1^2 + ξ2^2)} //
  FullSimplify
```

Out[333]=

```
{0, 0, 0}
```

In[334]:=

```
Tf = ((CoefficientArrays[{(TU /. x3 → 0), (TU /. x3 → h)} // Flatten // Simplify,
          Table[cj, {j, 1, 6}]][[2]] // Normal // Det) /.
      {ξ1 → m1, ξ2 → m2, m2^2 → R - m1^2, -m2^2 → -R + m1^2}) // Simplify;
```

In[335]:=

```
Tf1 = (Tf /. {m2^2 → R - m1^2, -m2^2 → -R + m1^2, m2^4 → R^2 - 2 R m1^2 + m1^4}) //
   Simplify
```

Out[335]=

$$-\frac{1}{\lambda+2\mu}\, e^{-i h\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+2\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)}\left(-1+e^{2 i h\sqrt{-R+\frac{\Lambda}{\mu}}}\right) R^2 \sqrt{-R+\frac{\Lambda}{\mu}}\;\mu^2$$

$$\Bigg(-32\, e^{2 i h\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)} R\,\mu\,(\lambda+2\mu)\,(\Lambda-2R\mu)^2\,(-\Lambda+R\mu)\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\; +$$

$$e^{i h\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)}\Bigg(\lambda\Bigg(-\Lambda^4\sqrt{-R+\frac{\Lambda}{\mu}}+16R^3\Lambda\mu^3\left(3\sqrt{-R+\frac{\Lambda}{\mu}}-4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+$$

$$8R\Lambda^3\mu\left(\sqrt{-R+\frac{\Lambda}{\mu}}-\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+32R^4\mu^4\left(-\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+$$

$$8R^2\Lambda^2\mu^2\left(-3\sqrt{-R+\frac{\Lambda}{\mu}}+5\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\Bigg)-$$

$$2\mu\,(-\Lambda+2R\mu)\Bigg(-\Lambda^3\sqrt{-R+\frac{\Lambda}{\mu}}+2R\Lambda^2\mu\left(3\sqrt{-R+\frac{\Lambda}{\mu}}-4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+16R^3\mu^3$$

$$\left(\sqrt{-R+\frac{\Lambda}{\mu}}-\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+4R^2\Lambda\mu^2\left(-5\sqrt{-R+\frac{\Lambda}{\mu}}+6\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\Bigg)\Bigg)+$$

$$e^{3 i h\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)}\Bigg(\lambda\Bigg(-\Lambda^4\sqrt{-R+\frac{\Lambda}{\mu}}+16R^3\Lambda\mu^3\left(3\sqrt{-R+\frac{\Lambda}{\mu}}-4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+$$

$$8R\Lambda^3\mu\left(\sqrt{-R+\frac{\Lambda}{\mu}}-\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+32R^4\mu^4\left(-\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+$$

$$8R^2\Lambda^2\mu^2\left(-3\sqrt{-R+\frac{\Lambda}{\mu}}+5\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\Bigg)-$$

$$2\mu\,(-\Lambda+2R\mu)\Bigg(-\Lambda^3\sqrt{-R+\frac{\Lambda}{\mu}}+2R\Lambda^2\mu\left(3\sqrt{-R+\frac{\Lambda}{\mu}}-4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+16R^3\mu^3$$

$$\left(\sqrt{-R+\frac{\Lambda}{\mu}}-\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+4R^2\Lambda\mu^2\left(-5\sqrt{-R+\frac{\Lambda}{\mu}}+6\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\Bigg)\Bigg)+$$

$$e^{i h\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)}\Bigg(\lambda\Bigg(\Lambda^4\sqrt{-R+\frac{\Lambda}{\mu}}-8R\Lambda^3\mu\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+$$

$$32R^4\mu^4\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)-16R^3\Lambda\mu^3$$

$$\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+8R^2\Lambda^2\mu^2\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+5\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\Bigg)+$$

$$2\mu(\Lambda-2R\mu)\left(\Lambda^3\sqrt{-R+\frac{\Lambda}{\mu}}-16R^3\mu^3\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)-\right.$$

$$2R\Lambda^2\mu\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+$$

$$\left.4R^2\Lambda\mu^2\left(5\sqrt{-R+\frac{\Lambda}{\mu}}+6\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\right)\Bigg)\Bigg)+$$

$$e^{i h\left(\sqrt{-R+\frac{\Lambda}{\mu}}+3\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)}\left(\lambda\left(\Lambda^4\sqrt{-R+\frac{\Lambda}{\mu}}-8R\Lambda^3\mu\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+\right.\right.$$

$$32R^4\mu^4\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)-16R^3\Lambda\mu^3$$

$$\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+8R^2\Lambda^2\mu^2\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+5\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\Bigg)+$$

$$2\mu(\Lambda-2R\mu)\left(\Lambda^3\sqrt{-R+\frac{\Lambda}{\mu}}-16R^3\mu^3\left(\sqrt{-R+\frac{\Lambda}{\mu}}+\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)-2R\Lambda^2\mu\right.$$

$$\left.\left.\left(3\sqrt{-R+\frac{\Lambda}{\mu}}+4\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)+4R^2\Lambda\mu^2\left(5\sqrt{-R+\frac{\Lambda}{\mu}}+6\sqrt{-R+\frac{\Lambda}{\lambda+2\mu}}\right)\right)\right)\Bigg)$$

In[336]:=

```
mydetT1 =
 (Tf1 /. {Sqrt[-R + Λ/(λ + 2 μ)] → Sqrt[C1], Sqrt[Λ - R μ]/Sqrt[μ] → Sqrt[C2], Sqrt[-R + Λ/μ] → Sqrt[C2],
    Sqrt[Λ - R μ] → Sqrt[C2] Sqrt[μ]}) // Simplify
```

Out[336]=

$$-\frac{1}{\lambda+2\mu}\sqrt{C2}\,e^{-i\left(2\sqrt{C1}+3\sqrt{C2}\right)h}\left(-1+e^{2i\sqrt{C2}h}\right)$$

$$R^2\mu^2\Big(-32\sqrt{C1}\,e^{2i\left(\sqrt{C1}+\sqrt{C2}\right)h}R\mu(\lambda+2\mu)(\Lambda-2R\mu)^2(-\Lambda+R\mu)+$$

$$e^{i\left(\sqrt{C1}+\sqrt{C2}\right)h}\Big(-8\sqrt{C1}\,R\mu(\lambda+2\mu)(\Lambda-2R\mu)^2(\Lambda-R\mu)-$$

$$\sqrt{C2}\Big(2\mu\left(\Lambda^4-8R\Lambda^3\mu+32R^2\Lambda^2\mu^2-56R^3\Lambda\mu^3+32R^4\mu^4\right)+$$

$$\lambda\left(\Lambda^4-8R\Lambda^3\mu+24R^2\Lambda^2\mu^2-48R^3\Lambda\mu^3+32R^4\mu^4\right)\Big)\Big)+$$

$$e^{3i\left(\sqrt{C1}+\sqrt{C2}\right)h}\Big(-8\sqrt{C1}\,R\mu(\lambda+2\mu)(\Lambda-2R\mu)^2(\Lambda-R\mu)-$$

$$\sqrt{C2}\Big(2\mu\left(\Lambda^4-8R\Lambda^3\mu+32R^2\Lambda^2\mu^2-56R^3\Lambda\mu^3+32R^4\mu^4\right)+$$

$$\lambda\left(\Lambda^4-8R\Lambda^3\mu+24R^2\Lambda^2\mu^2-48R^3\Lambda\mu^3+32R^4\mu^4\right)\Big)\Big)+$$

$$e^{i\left(3\sqrt{C1}+\sqrt{C2}\right)h}\Big(-8\sqrt{C1}\,R\mu(\lambda+2\mu)(\Lambda-2R\mu)^2(\Lambda-R\mu)+$$

$$\sqrt{C2}\Big(2\mu\left(\Lambda^4-8R\Lambda^3\mu+32R^2\Lambda^2\mu^2-56R^3\Lambda\mu^3+32R^4\mu^4\right)+$$

$$\lambda\left(\Lambda^4-8R\Lambda^3\mu+24R^2\Lambda^2\mu^2-48R^3\Lambda\mu^3+32R^4\mu^4\right)\Big)\Big)+$$

$$e^{i\left(\sqrt{C1}+3\sqrt{C2}\right)h}\Big(-8\sqrt{C1}\,R\mu(\lambda+2\mu)(\Lambda-2R\mu)^2(\Lambda-R\mu)+$$

$$\sqrt{C2}\Big(2\mu\left(\Lambda^4-8R\Lambda^3\mu+32R^2\Lambda^2\mu^2-56R^3\Lambda\mu^3+32R^4\mu^4\right)+$$

$$\lambda\left(\Lambda^4-8R\Lambda^3\mu+24R^2\Lambda^2\mu^2-48R^3\Lambda\mu^3+32R^4\mu^4\right)\Big)\Big)\Big)$$

In[337]:=

```
mydetT2 = (mydetT1 // ExpToTrig // TrigExpand // Factor)
```

Out[337]=

$$
\begin{aligned}
&\frac{1}{\lambda+2\mu}\,4\,i\,\sqrt{C2}\,R^2\,\mu^2\,\Big(-\sqrt{C2}\,\lambda\,\Lambda^4\,\text{Sin}[\sqrt{C1}\,h]+8\,\sqrt{C2}\,R\,\lambda\,\Lambda^3\,\mu\,\text{Sin}[\sqrt{C1}\,h]-\\
&\quad 2\,\sqrt{C2}\,\Lambda^4\,\mu\,\text{Sin}[\sqrt{C1}\,h]-24\,\sqrt{C2}\,R^2\,\lambda\,\Lambda^2\,\mu^2\,\text{Sin}[\sqrt{C1}\,h]+16\,\sqrt{C2}\,R\,\Lambda^3\,\mu^2\,\text{Sin}[\sqrt{C1}\,h]+\\
&\quad 48\,\sqrt{C2}\,R^3\,\lambda\,\Lambda\,\mu^3\,\text{Sin}[\sqrt{C1}\,h]-64\,\sqrt{C2}\,R^2\,\Lambda^2\,\mu^3\,\text{Sin}[\sqrt{C1}\,h]-\\
&\quad 32\,\sqrt{C2}\,R^4\,\lambda\,\mu^4\,\text{Sin}[\sqrt{C1}\,h]+112\,\sqrt{C2}\,R^3\,\Lambda\,\mu^4\,\text{Sin}[\sqrt{C1}\,h]-64\,\sqrt{C2}\,R^4\,\mu^5\,\text{Sin}[\sqrt{C1}\,h]+\\
&\quad \sqrt{C2}\,\lambda\,\Lambda^4\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]-8\,\sqrt{C2}\,R\,\lambda\,\Lambda^3\,\mu\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]+\\
&\quad 2\,\sqrt{C2}\,\Lambda^4\,\mu\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]+24\,\sqrt{C2}\,R^2\,\lambda\,\Lambda^2\,\mu^2\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]-\\
&\quad 16\,\sqrt{C2}\,R\,\Lambda^3\,\mu^2\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]-48\,\sqrt{C2}\,R^3\,\lambda\,\Lambda\,\mu^3\\
&\quad\ \text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]+64\,\sqrt{C2}\,R^2\,\Lambda^2\,\mu^3\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]+\\
&\quad 32\,\sqrt{C2}\,R^4\,\lambda\,\mu^4\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]-112\,\sqrt{C2}\,R^3\,\Lambda\,\mu^4\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]+\\
&\quad 64\,\sqrt{C2}\,R^4\,\mu^5\,\text{Cos}[\sqrt{C2}\,h]^2\,\text{Sin}[\sqrt{C1}\,h]-16\,\sqrt{C1}\,R\,\lambda\,\Lambda^3\,\mu\,\text{Sin}[\sqrt{C2}\,h]+\\
&\quad 80\,\sqrt{C1}\,R^2\,\lambda\,\Lambda^2\,\mu^2\,\text{Sin}[\sqrt{C2}\,h]-32\,\sqrt{C1}\,R\,\Lambda^3\,\mu^2\,\text{Sin}[\sqrt{C2}\,h]-\\
&\quad 128\,\sqrt{C1}\,R^3\,\lambda\,\Lambda\,\mu^3\,\text{Sin}[\sqrt{C2}\,h]+160\,\sqrt{C1}\,R^2\,\Lambda^2\,\mu^3\,\text{Sin}[\sqrt{C2}\,h]+\\
&\quad 64\,\sqrt{C1}\,R^4\,\lambda\,\mu^4\,\text{Sin}[\sqrt{C2}\,h]-256\,\sqrt{C1}\,R^3\,\Lambda\,\mu^4\,\text{Sin}[\sqrt{C2}\,h]+\\
&\quad 128\,\sqrt{C1}\,R^4\,\mu^5\,\text{Sin}[\sqrt{C2}\,h]+16\,\sqrt{C1}\,R\,\lambda\,\Lambda^3\,\mu\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]-\\
&\quad 80\,\sqrt{C1}\,R^2\,\lambda\,\Lambda^2\,\mu^2\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]+\\
&\quad 32\,\sqrt{C1}\,R\,\Lambda^3\,\mu^2\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]+\\
&\quad 128\,\sqrt{C1}\,R^3\,\lambda\,\Lambda\,\mu^3\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]-\\
&\quad 160\,\sqrt{C1}\,R^2\,\Lambda^2\,\mu^3\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]-\\
&\quad 64\,\sqrt{C1}\,R^4\,\lambda\,\mu^4\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]+\\
&\quad 256\,\sqrt{C1}\,R^3\,\Lambda\,\mu^4\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]-\\
&\quad 128\,\sqrt{C1}\,R^4\,\mu^5\,\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\,\text{Sin}[\sqrt{C2}\,h]-\\
&\quad \sqrt{C2}\,\lambda\,\Lambda^4\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2+8\,\sqrt{C2}\,R\,\lambda\,\Lambda^3\,\mu\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2-\\
&\quad 2\,\sqrt{C2}\,\Lambda^4\,\mu\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2-24\,\sqrt{C2}\,R^2\,\lambda\,\Lambda^2\,\mu^2\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2+\\
&\quad 16\,\sqrt{C2}\,R\,\Lambda^3\,\mu^2\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2+\\
&\quad 48\,\sqrt{C2}\,R^3\,\lambda\,\Lambda\,\mu^3\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2-\\
&\quad 64\,\sqrt{C2}\,R^2\,\Lambda^2\,\mu^3\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2-32\,\sqrt{C2}\,R^4\,\lambda\,\mu^4\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2+\\
&\quad 112\,\sqrt{C2}\,R^3\,\Lambda\,\mu^4\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2-64\,\sqrt{C2}\,R^4\,\mu^5\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]^2\Big)
\end{aligned}
$$

In[338]:=

```
mydetT3 = mydetT2 / (1/(λ + 2 μ) 4 ⅈ Sqrt[C2] R^2 μ^2) // FullSimplify
```

Out[338]=

$$
\begin{aligned}
&2\,\text{Sin}[\sqrt{C2}\,h]\,\Big(8\,\sqrt{C1}\,R\,\mu\,(\lambda+2\,\mu)\,(\Lambda-2\,R\,\mu)^2\,(\Lambda-R\,\mu)\,\big(-1+\text{Cos}[\sqrt{C1}\,h]\,\text{Cos}[\sqrt{C2}\,h]\big)-\\
&\quad \sqrt{C2}\,\big(\lambda\,\Lambda^4+2\,\Lambda^3\,(-4\,R\,\lambda+\Lambda)\,\mu+8\,R\,(3\,R\,\lambda-2\,\Lambda)\,\Lambda^2\,\mu^2+16\,R^2\,\Lambda\,(-3\,R\,\lambda+4\,\Lambda)\,\mu^3+\\
&\qquad 16\,R^3\,(2\,R\,\lambda-7\,\Lambda)\,\mu^4+64\,R^4\,\mu^5\big)\,\text{Sin}[\sqrt{C1}\,h]\,\text{Sin}[\sqrt{C2}\,h]\Big)
\end{aligned}
$$

In[339]:=

```
mydetT3 /. {C2 → -AC2, C1 → -AC1} /.
  {Sqrt[-AC2] → I Sqrt[AC2], Sqrt[-AC1] → I Sqrt[AC1]} // Simplify
mydetT3 /. { C1 → -AC1, C2 → AC2} /. {Sqrt[-AC1] → I Sqrt[AC1]} // Factor //
 Simplify
```

Out[339]=

$$2\,\mathrm{Sinh}\left[\sqrt{AC2}\,h\right]\left(-8\sqrt{AC1}\,R\,\mu\,(\lambda+2\mu)\,(\Lambda-2R\mu)^2\,(\Lambda-R\mu)\,\left(-1+\mathrm{Cosh}\left[\sqrt{AC1}\,h\right]\mathrm{Cosh}\left[\sqrt{AC2}\,h\right]\right)-\sqrt{AC2}\,\left(\lambda\Lambda^4+2\Lambda^3\,(-4R\lambda+\Lambda)\,\mu+8R\,(3R\lambda-2\Lambda)\,\Lambda^2\mu^2+16R^2\Lambda\,(-3R\lambda+4\Lambda)\,\mu^3+16R^3\,(2R\lambda-7\Lambda)\,\mu^4+64R^4\mu^5\right)\mathrm{Sinh}\left[\sqrt{AC1}\,h\right]\mathrm{Sinh}\left[\sqrt{AC2}\,h\right]\right)$$

Out[340]=

$$-2\,i\,\mathrm{Sin}\left[\sqrt{AC2}\,h\right]\left(8\sqrt{AC1}\,R\,\mu\,(\lambda+2\mu)\,(\Lambda-2R\mu)^2\,(\Lambda-R\mu)+8\sqrt{AC1}\,R\,\mu\,(\lambda+2\mu)\,(\Lambda-2R\mu)^2\,(-\Lambda+R\mu)\,\mathrm{Cos}\left[\sqrt{AC2}\,h\right]\mathrm{Cosh}\left[\sqrt{AC1}\,h\right]+\sqrt{AC2}\,\left(2\mu\left(\Lambda^4-8R\Lambda^3\mu+32R^2\Lambda^2\mu^2-56R^3\Lambda\mu^3+32R^4\mu^4\right)+\lambda\left(\Lambda^4-8R\Lambda^3\mu+24R^2\Lambda^2\mu^2-48R^3\Lambda\mu^3+32R^4\mu^4\right)\right)\mathrm{Sin}\left[\sqrt{AC2}\,h\right]\mathrm{Sinh}\left[\sqrt{AC1}\,h\right]\right)$$

In[341]:=

```
MyDetTSign1[λ_, μ_, h_, R_ , Λ_] :=
  (* for Λ < R μ *) With[{AC1 = R - Λ/(λ + 2 μ), AC2 = -Λ / μ + R},
   (-8 Sqrt[AC1] R μ (λ + 2 μ) (Λ - 2 R μ)^2 (Λ - R μ) (-1 + Cosh[Sqrt[AC1] h] Cosh[Sqrt[AC2] h]) -
      Sqrt[AC2] (λ Λ^4 + 2 Λ^3 (-4 R λ + Λ) μ + 8 R (3 R λ - 2 Λ) Λ^2 μ^2 +
          16 R^2 Λ (-3 R λ + 4 Λ) μ^3 + 16 R^3 (2 R λ - 7 Λ) μ^4 + 64 R^4 μ^5)
        Sinh[Sqrt[AC1] h] Sinh[Sqrt[AC2] h]) / (Cosh[h Sqrt[AC1]] Cosh[h Sqrt[AC2]])];
MyDetTSign2Part1[λ_, μ_, h_, R_ , Λ_] :=  (* for R μ < Λ < R(λ+2 μ)   *)
  With[{AC1 = R - Λ/(λ + 2 μ), AC2 = Λ / μ - R}, Sin[Sqrt[AC2] h]];
MyDetTSign2Part2[λ_, μ_, h_, R_ , Λ_] :=  (* for R μ < Λ < R(λ+2 μ)   *)
  With[{AC1 = R - Λ/(λ + 2 μ), AC2 = Λ / μ - R}, (8 Sqrt[AC1] R μ (λ + 2 μ) (Λ - 2 R μ)^2 (Λ - R μ) +
      8 Sqrt[AC1] R μ (λ + 2 μ) (Λ - 2 R μ)^2 (-Λ + R μ) Cos[Sqrt[AC2] h] Cosh[Sqrt[AC1] h] +
      Sqrt[AC2] (2 μ (Λ^4 - 8 R Λ^3 μ + 32 R^2 Λ^2 μ^2 - 56 R^3 Λ μ^3 + 32 R^4 μ^4) +
         λ (Λ^4 - 8 R Λ^3 μ + 24 R^2 Λ^2 μ^2 - 48 R^3 Λ μ^3 + 32 R^4 μ^4))
       Sin[Sqrt[AC2] h] Sinh[Sqrt[AC1] h]) / (Cosh[h Sqrt[AC1]] Λ)^2];
MyDetTSign3Part1[λ_, μ_, h_, R_ , Λ_] :=  (* for Λ > R(λ+2 μ)   *)
  With[{C1 = -R + Λ/(λ + 2 μ), C2 = Λ / μ - R}, 2 Sin[Sqrt[C2] h]];
MyDetTSign3Part2[λ_, μ_, h_, R_ , Λ_] :=
  (* for Λ > R(λ+2 μ)   *) With[{C1 = -R + Λ/(λ + 2 μ), C2 = Λ / μ - R},
    (8 Sqrt[C1] R μ (λ + 2 μ) (Λ - 2 R μ)^2 (Λ - R μ) (-1 + Cos[Sqrt[C1] h] Cos[Sqrt[C2] h]) -
      Sqrt[C2] (λ Λ^4 + 2 Λ^3 (-4 R λ + Λ) μ + 8 R (3 R λ - 2 Λ) Λ^2 μ^2 + 16 R^2 Λ (-3 R λ + 4 Λ) μ^3 +
         16 R^3 (2 R λ - 7 Λ) μ^4 + 64 R^4 μ^5) Sin[Sqrt[C1] h] Sin[Sqrt[C2] h]) ] / Λ^4;
```

In[356]:=

```
(* numerically computed eigenvalues *)
MyTEigs[λ_, μ_, h_, ΛMax_] := Module[{evs, R, Rmax, n, Rsq,
    solns1, solns21, solns22, solns31, solns32, j, k, i6, M, M2},
   PrintTemporary["λ=", λ, " μ=", μ];
   evs = Table[{0, 1, N[(λ + 2 μ) Pi^2 n^2 / h^2]},
     {n, 1, h Sqrt[ΛMax] / (Pi Sqrt[λ + 2 μ])}];
   (* the eigenvalues above come from R=0, with multiplicity one *)
   evs = AppendTo[evs, {0, 4, 0}]; (*ev=0*)
   Monitor[Do[Rsq = SquaresR[2, R];
     If[Rsq ≠ 0,
      M = Min[μ R, ΛMax] - 10^(-6);
      solns1 = findAllRoots[
        MyDetTSign1[λ, μ, h, R, Λ] / R^3, {Λ, 10^(-4), M}, PlotPoints → 10 000];
      Do[AppendTo[evs, {R, Rsq, solns1〚j〛}], {j, 1, Length[solns1]}];
      AppendTo[evs, {R, Rsq, μ R}];
      If[R ≤ ΛMax / μ,
       M2 = Min[(λ + 2 μ) R, ΛMax] - 10^(-6);
       solns21 = findAllRoots[MyDetTSign2Part1[λ, μ, h, R, Λ],
         {Λ, μ R + 10^(-4), M2}, PlotPoints → 10 000];
       solns22 = findAllRoots[MyDetTSign2Part2[λ, μ, h, R, Λ],
         {Λ, μ R + 10^(-4), M2}, PlotPoints → 10 000];
       Do[AppendTo[evs, {R, Rsq, solns21〚j〛}], {j, 1, Length[solns21]}];
       Do[AppendTo[evs, {R, Rsq, solns22〚j〛}], {j, 1, Length[solns22]}];
       solns31 = findAllRoots[MyDetTSign3Part1[λ, μ, h, R, Λ],
         {Λ, (λ + 2 μ) R + 10^(-4), ΛMax}, PlotPoints → 10 000];
       solns32 = findAllRoots[MyDetTSign3Part2[λ, μ, h, R, Λ],
         {Λ, (λ + 2 μ) R + 10^(-4), ΛMax}, PlotPoints → 10 000];
       Do[AppendTo[evs, {R, Rsq, solns31〚j〛}], {j, 1, Length[solns31]}];
       Do[AppendTo[evs, {R, Rsq, solns32〚j〛}], {j, 1, Length[solns32]}];
      ];
     ]
     , {R, 1, 3 / 2 ΛMax / μ}], ProgressIndicator[R, {1, 3 / 2 ΛMax / μ}]];
   SortBy[evs, N[#〚3〛] &]
  ];
```

In[347]:=

```
λs = {-1 / 2, 3, 100}; hs = {1 / 2, 2}; Lmax = {1000, 500, 250};
htext = {"0.5", "2"}; ylim = {5000, 2500, 1200};
lambdatext = {"-0.5", "3", "100"};
xticks = Table[
   {{Lmax〚j〛 / 2, MaTeX[Lmax〚j〛 / 2]}, {Lmax〚j〛, MaTeX[Lmax〚j〛]}}, {j, 1, 3}] ;
yticks = Table[
   {{ylim〚i〛 / 2, MaTeX[ylim〚i〛 / 2]}, {ylim〚i〛, MaTeX[ylim〚i〛]}}, {i, 1, 2}] ;
```

In[ ]:=

```
(* EVTs=Table[MyTEigs[λs〚j〛,1,hs〚i〛,Lmax〚i〛],{i,1,2},{j,1,3}]; *)
```

In[362]:=

```
Lmax1 = {300, 400, 500};
EVTs1 = Table[MyTEigs[λs⟦j⟧, 1, 1, 1.1 Lmax1⟦j⟧], {j, 1, 3}];
```

FindRoot: Failed to converge to the requested accuracy or precision within 100 iterations.

FindRoot: Failed to converge to the requested accuracy or precision within 100 iterations.

FindRoot: Failed to converge to the requested accuracy or precision within 100 iterations.

General: Further output of FindRoot::cvmit will be suppressed during this calculation.

In[ ]:=

```
(* graphicstableFree=
 Table[
  Plot[{MyNlambda3d[EVTs⟦i⟧⟦j⟧,Λ]-CWeyl[3,λs⟦j⟧,1]Vol3[hs⟦i⟧]  Λ^(3/2),
    Bfree[3,αλ[λs⟦j⟧,1],1]Area3[hs⟦i⟧]Λ},
   {Λ, 0, Lmax⟦i⟧}, PlotStyle→{Black,Blue}, Exclusions→None,
   Axes→True,
   AxesLabel→{MaTeX["\\Lambda"], None},
   PlotRange→{{0,Lmax⟦i⟧},{0,ylim⟦i⟧}},
   Ticks→{xticks⟦i⟧,yticks⟦i⟧},
   Epilog→Inset[Framed[MaTeX["\\lambda="<>lambdatext⟦j⟧<>", h="<>htext⟦i⟧],
      RoundingRadius→2, ContentPadding→False,FrameMargins→Tiny],
     Scaled[{0.1,1}],Scaled[{0,1}]]
  ],{i,1,2},{j,1,3}]  *)
```

In[353]:=

```
xticks1 = Table[{{Lmax1⟦j⟧ / 2, MaTeX[Lmax1⟦j⟧ / 2]},
    {Lmax1⟦j⟧, MaTeX[Lmax1⟦j⟧]}}, {j, 1, 3}] ;
yticks1 = {{1000, MaTeX[1000]}, {2000, MaTeX[2000]}};
```

In[364]:=

```
graphicstable1Free =
 Table[
  Plot[{MyNlambda3d[EVTs1⟦j⟧, Λ] - CWeyl[3, λs⟦j⟧, 1] × Vol3[1] Λ^(3 / 2),
    Bfree[3, αλ[λs⟦j⟧, 1], 1] × Area3[1] Λ},
   {Λ, 0, Lmax1⟦j⟧}, PlotStyle → {Black, Blue}, Exclusions → None,
   Axes → True,
   AxesLabel → {MaTeX["\\Lambda"], None},
   PlotRange → {{0, Lmax1⟦j⟧}, Automatic},
   Ticks → {xticks1⟦j⟧, yticks1},
   Epilog → Inset[Framed[MaTeX["\\lambda=" <> lambdatext⟦j⟧ <> ",\\quad h=1"],
      RoundingRadius → 2, ContentPadding → False,
      FrameMargins → Tiny], Scaled[{0.1, 1}], Scaled[{0, 1}]]
  ], {j, 1, 3}]
```

Out[364]=



In[365]:=

```
legendcylfree = LineLegend[{Black, Blue},
  MaTeX[{"\\text{computed }\\mathcal{N}_{\\mathrm{free}}(\\Lambda)-
      a (2\\pi)^2 h \\Lambda^{3/2}\\qquad\\qquad",
    "8\\pi^2 b_{\\mathrm{free}}\\Lambda"}], LegendFunction →
   None, LegendLayout → {"Row", 1}]
```

Out[365]=

computed $\mathcal{N}_{\mathrm{free}}(\Lambda) - a(2\pi)^2 h\Lambda^{3/2}$ $8\pi^2 b_{\mathrm{free}}\Lambda$

In[366]:=

```
graphicstable1Freenew =
 Table[
  Plot[{MyNlambda3d[EVTs1[[j]], Λ] - CWeyl[3, λs[[j]], 1] × Vol3[1] Λ^(3/2),
    Bfree[3, αλ[λs[[j]], 1], 1] × Area3[1] Λ, -BDirLiu[3, λs[[j]], 1] × Area3[1] Λ},
   {Λ, 0, Lmax1[[j]]}, PlotStyle → {Black, Blue, Red}, Exclusions → None,
   Axes → True,
   AxesLabel → {MaTeX["\\Lambda"], None},
   PlotRange → {{0, Lmax1[[j]]}, Automatic},
   Ticks → {xticks1[[j]], yticks1},
   Epilog → Inset[Framed[MaTeX["\\lambda=" <> lambdatext[[j]] <> ",\\quad h=1"],
      RoundingRadius → 2, ContentPadding → False,
      FrameMargins → Tiny], Scaled[{0.1, 1}], Scaled[{0, 1}]]
  ], {j, 1, 3}]
```

Out[366]=



In[367]:=

```
legendcylfreenew = LineLegend[{Black, Blue, Red}, MaTeX[
   {"\\text{computed }\\mathcal{N}_{\\mathrm{free}}(\\Lambda)- a (2\\pi)^2 h
      \\Lambda^{3/2}\\qquad\\qquad", "8\\pi^2 b_{\\mathrm{free}}\\Lambda",
    "8\\pi^2 b_{\\mathrm{free}}^{\\mathrm{Liu}}\\Lambda"}],
  LegendFunction → None, LegendLayout → {"Row", 1}]
```

Out[367]=

computed $\mathcal{N}_{\mathrm{free}}(\Lambda) - a(2\pi)^2 h\Lambda^{3/2}$ $8\pi^2 b_{\mathrm{free}}\Lambda$ $8\pi^2 b_{\mathrm{free}}^{\mathrm{Liu}}\Lambda$

In[368]:=

```
figcylfree = GraphicsColumn[{GraphicsRow[graphicstable1Free,
    ImageSize → Full, Spacings → {Scaled[0.05], Automatic}],
   legendcylfree}, ImageSize → Full, Spacings → {0, Automatic}]
```

Out[368]=



In[∘]:=

```
Export["figcylfree.pdf", figcylfree]
```

Out[∘]=

```
figcylfree.pdf
```

In[369]:=

```
figcylfreenew = GraphicsColumn[{GraphicsRow[graphicstable1Freenew,
    ImageSize → Full, Spacings → {Scaled[0.05], Automatic}],
   legendcylfreenew}, ImageSize → Full, Spacings → {0, Automatic}]
```

Out[369]=



In[∘]:=

```
Export["figcylfreenew.pdf", figcylfreenew]
```

Out[∘]=

```
figcylfreenew.pdf
```